



=====SIMPLE_AM_WAVEFORM_GENERATION=====

- 1) 100%_AM_1V_pk (This Spice version) = 1_carrier_pk and 0.5_sidebands_pk
- 2) Signal_rms = sqrt(2)*Signal_pk
- 3) 100%_AM_1V_pk (Spectrum RMS) = sqrt(1+.5^2+.5^2)/sqrt(2)_V_rms
- 4) FFT_resolution_Hz = 1/total_time_sec
- 5) Nyquist_Hz = 0.5/Sample_time_sec
- 6) Num_Bins for FFT = Nyquist_Hz/FFT_resolution_Hz



- 1) For this version of spice, 100%_AM_1V_pk signal produces a carrier signal with a magnitude of one with two 0.5 magnitude sidebands.
- 2) Note the rms value for a 1V_pk signal is .707V_rms, which can be calculated in spice.

```
=====Find Ave RmsCOS=====
RMS_level_Expect .707 RMS_level_Cos 0.707494
=====Find Ave RmsAM=====
RMS_level_Expect .866 RMS_level_RM 0.86555
=====FFT and Plot_AM=====
FFT_BandWidth_Hz= 500000
FFT_resolution_Hz= 1000
Fundamental+sideband 0.999954 + 0.499976 + 0.499978
Total_RMS 1.22469
=====FFT and Plot_Vcos=====
FFT_BandWidth_Hz= 500000
FFT_resolution_Hz= 1000
Fundamental+sideband 0.999954 + 6.25736E-07 + 5.03318E-07
Total_RMS 0.999954
```

- 3) The spectrum's rms value for a 100%_AM 1V_pk signal calculates to be 1.225V_rms.
- 4) One needs for instance 1 whole second to measure a signal to a 1Hz resolution.
- 5) At least two samples are needed to detect the presents of an AC signal. That is why Nyquist is half the sample rate.
- 6) FFT translate samples over time to bins of frequency. The bin's width is the FFT_resolution_Hz and the Maximum frequency is Nyquist_Hz

=====MacSpiceCode=====

```
SIMPLE_AM_WAVEFORM_GENERATION
*=====Create Signal=====
VTime VTime 0 DC 0 PWL( 0 0 1 1)
Vfreq1 Vfreq1 0 DC 2
BMOD VMOD 0 V = cos(6.2831853*2000*V(VTime))
BAM VAM 0 V = (1-V(VMOD))*cos(6.2831853*100k*V(VTime))
BCOS VCOS 0 V = 1*cos(6.2831853*100k*V(VTime))
```

```
.control
*TRAN TSTEP TSTOP TSTART TMAX ?UIC?
tran .1u 1m 0 .1u
set pensize = 2
```

```

plot          vam vcos
echo          "=====Find_Ave_RmsCOS=====
let averVal = mean(vcos)
let noisAC = vcos - averVal
let RmsVal = sqrt(mean(noisAC* noisAC))
echo          "RMS_level_Expect      .707      RMS_level_Cos      $&RmsVal      "
unlet averVal
unlet RmsVal
echo          "=====Find_Ave_RmsAM=====
let averVal = mean(VAM)
let noisAC = VAM - averVal
let RmsVal = sqrt(mean(noisAC* noisAC))
echo          "RMS_level_Expect      .866      RMS_level_RM      $&RmsVal      "
unlet averVal
unlet RmsVal
echo          "=====FFT_and_Plot_AM=====
linearize
let          FFT_BandWidth_Hz =      500k
let          FFT_resolution_Hz =      1k
echo          "FFT_BandWidth_Hz=      $&FFT_BandWidth_Hz"
echo          "FFT_resolution_Hz=      $&FFT_resolution_Hz"
set          specwindow=      "rectangular"
spec          $&FFT_resolution_Hz      $&FFT_BandWidth_Hz      $&FFT_resolution_Hz      v(vam)
plot          mag (vam) ylog xlimit 90k 110k
plot          mag (vam) ylog xlimit 95k 105k ylimit .1 1
let fund = mag(vam[99])
let upsb = mag(vam[101])
let lpsb = mag(vam[97])
echo          "Fundamental+sideband      $&fund + $&upsb + $&lpsb      "
let totalrms = sqrt( fund*fund +upsb*upsb+ lpsb*lpsb)
echo          "Total_RMS      $&totalrms      "
echo          "=====FFT_and_Plot_Vcos=====
destroy
let          FFT_BandWidth_Hz =      500k
let          FFT_resolution_Hz =      1k
echo          "FFT_BandWidth_Hz=      $&FFT_BandWidth_Hz"
echo          "FFT_resolution_Hz=      $&FFT_resolution_Hz"
set          specwindow=      "rectangular"
spec          $&FFT_resolution_Hz      $&FFT_BandWidth_Hz      $&FFT_resolution_Hz      v(vcos)
plot          mag (vcos) ylog xlimit 90k 110k
plot          mag (vcos) ylog xlimit 95k 105k ylimit .1 1
let fund = mag(vcos[99])
let upsb = mag(vcos[101])
let lpsb = mag(vcos[97])
echo          "Fundamental+sideband      $&fund + $&upsb + $&lpsb      "
let totalrms = sqrt( fund*fund +upsb*upsb+ lpsb*lpsb)
echo          "Total_RMS      $&totalrms      "

.endc
.end

```

4.4.11_11.55AM
dsauersanjose@aol.com
Don Sauer